

Instructor: Matt Fredrikson

TA: Ryan Wagner

Incremental Proof Development in Dafny

These notes continue the discussion from Lecture 17 on proving the correctness of imperative code in Dafny. Before we focused on partial correctness, and didn't worry about termination. Here, we'll focus on termination. We'll start with the termination argument for `BubbleSort`, for which we proved partial correctness before. We'll then take a look at a different method that naturally requires a different kind of well-founded relation to prove termination.

Proving Termination

After proving that the method satisfies its postcondition whenever it terminates, in order to prove total correctness we must prove that it always terminates when the preconditions are satisfied. Notice that we did not annotate our program with any ranking functions. Dafny was able to figure out the appropriate ranking functions for our code, and insert them automatically. In this case, the following ranking functions would work well.

```
1 method BubbleSort(a: array<int>)
2   modifies a
3   requires a != null
4   {
5     var i := a.Length - 1;
6     while(i > 0)
7       invariant 0 <= i+1
8       decreases i+1, i+1
9     {
10      var j := 0;
11      while(j < i)
12        invariant 0 <= i+1 && 0 <= i - j
13        decreases i+1, i-j
14      {
15        if(a[j] > a[j+1])
16        {
17          a[j], a[j+1] := a[j+1], a[j];
18        }
19        j := j + 1;
20      }
21      i := i - 1;
22    }
23  }
```

However, from the perspective of a programmer writing code in Dafny, one could also write:

```

1 method BubbleSort(a: array<int>)
2   modifies a
3   requires a != null
4   {
5     var i := a.Length - 1;
6     while(i > 0)
7       invariant -1 <= i
8       decreases i
9     {
10      var j := 0;
11      while(j < i)
12        invariant 0 <= i - j
13        decreases i-j
14      {
15        ...
16        j := j + 1;
17      }
18      i := i - 1;
19    }
20  }

```

We could drop the invariants as well, as Dafny is able to infer them without assistance. However, would these ranking functions work if we tried to prove them by hand? Consider the basic path leaving the inner loop, and going back to execute the outer loop again.

```

1 invariant 0 <= i - j
2 decreases i - j
3 assume j >= i;
4 i := i - 1;
5 decreases i

```

The corresponding verification condition is

$$\begin{aligned}
 & 0 \leq i - j \rightarrow \text{wlp}(\text{assume } j > i; i := i - 1, i < i' - j')[i/i', j/j'] \\
 \Leftrightarrow & 0 \leq i - j \wedge j \geq i \rightarrow i - 1 < i - j
 \end{aligned}$$

This is not valid, and in fact the consequent is only true when $i = j = 0$.

The simpler ranking functions are accepted by Dafny because it does not generate exactly the same verification conditions that we do when proving termination by hand. To better understand how Dafny constructs a termination argument, consider the following program.

```

1 method BubbleSort(a: array<int>)
2   modifies a
3   requires a != null
4   {
5     var i := a.Length - 1;
6     while(i > 0)
7       invariant -1 <= i
8       decreases i
9     {
10      ghost var rankval_0 := i;
11      var j := 0;
12      while(j < i)
13        invariant 0 <= i - j
14        decreases i-j
15      {
16        ghost var rankval_1 := i - j;
17        ...
18        j := j + 1;
19        assert i - j < rankval_1;
20      }
21      i := i - 1;
22      assert i < rankval_0;
23    }
24  }

```

Roughly, this corresponds to the checks performed by Dafny given these **ranking** annotations. We could prove this by hand as well, showing that the assertions always hold by generating additional verification conditions for the assertions. We would arrive at the same result as we do using the “standard” termination approach with the first set of ranking functions, but doing so would require developing additional invariants (e.g., that i is not modified by the inner loop), and would not necessarily lead to a simpler proof in the end.

When developing termination proofs by hand, you should use the uniform approach we’ve discussed in class rather than the method involving ghost state used by Dafny. When working with Dafny, you may find it helpful to realize that this is how it constructs its termination argument to gain insight into why your annotations are not being accepted by the verifier. However, if you construct a valid ranking function that you are able to prove using the standard method, then you should in most cases be able to use it as an annotation in your code that will be accepted by the verifier. If the verifier does not accept what appears to be a perfectly good ranking function, then you should proceed to derive the relevant verification conditions in this manner to check your work. Usually, you will be able to identify a path on which your function does not decrease, or in certain cases, a verification condition that looks as though it will be difficult for the decision procedure to automatically discharge (e.g., involving complex terms from the theory of integers or reals).

For completeness, let’s walk through the termination argument on the properly-annotated method (i.e., the first version of `BubbleSort` listed in this section). Notice that we do not have a ranking function on the method itself, as it is not recursive. Our termination argument will focus solely on the loops contained in the body, so we can ignore the basic paths that enter and exit the outer loop. We begin with the basic path starting at the top of the outer loop, entering the inner loop.

```

1 invariant 0 <= i+1
2 decreases i+1, i+1
3 assume i > 0;
4 j := 0;
5 decreases i+1, i-j

```

Our verification condition is given by:

$$\begin{aligned}
& 0 \leq i+1 \rightarrow \text{wlp}(\text{assume } i > 0; j := 0, (i+1, i-j) < (i'+1, i'+1))[i/i', j/j'] \\
\Leftrightarrow & 0 \leq i+1 \rightarrow \text{wlp}(\text{assume } i > 0, (i+1, i-0) < (i'+1, i'+1))[i/i', j/j'] \\
\Leftrightarrow & 0 \leq i+1 \rightarrow (i > 0 \rightarrow (i+1, i) < (i'+1, i'+1))[i/i', j/j'] \\
\Leftrightarrow & 0 \leq i+1 \rightarrow (i+1, i) < (i+1, i+1)
\end{aligned}$$

The last step removes the additional antecedent $i > 0$, because it is implied by $0 \leq i+1$, as well as substituting the primed variables. We see that it is valid because of the lexicographic ordering, as $i+1 = i+1$ and $i < i+1$.

The next basic path corresponds to a non-exiting iteration of the inner loop. The comments label the commands, so that we do not need to write them in full when constructing the verification condition.

```

1 invariant 0 <= i+1 && 0 <= i - j
2 decreases i+1, i-j
3 assume j < i; // c0
4 assume a[j] > a[j+1]; // c1
5 a[j], a[j+1] := a[j+1], a[j]; // c2
6 j := j + 1; // c3
7 decreases i+1, i-j

```

The verification condition is:

$$\begin{aligned}
& 0 \leq i+1 \wedge 0 \leq i-j \rightarrow \text{wlp}(c_0; c_1; c_2; c_3, (i+1, i-j) < (i'+1, i'-j'))[i/i', j/j'] \\
\Leftrightarrow & 0 < i \wedge 0 \leq i-j \wedge j < i \wedge a[j] > a[j+1] \rightarrow (i+1, i-(j+1)) < (i+1, i-j)
\end{aligned}$$

This is valid, because $i-(j+1) < i-j$. Notice that the statements involving the array have no bearing on the validity of this condition. In the interest of keeping the verification condition simpler, we could have just written:

$$0 < i \wedge 0 \leq i-j \wedge j < i \rightarrow (i+1, i-(j+1)) < (i+1, i-j)$$

However, when doing these proofs by hand, it's best to write the entire VC out at first, and simplify afterwards with an additional step that may help to clarify the reasoning. Moving onto the next basic path:

```

1 invariant 0 <= i+1 && 0 <= i - j
2 decreases i+1, i-j
3 assume j < i; // c0
4 assume a[j] <= a[j+1]; // c1
5 j := j + 1; // c2
6 decreases i+1, i-j

```

The verification condition is identical to the previous one with the exception of the condition in the **assume** statement, which is irrelevant to the termination argument:

$$\begin{aligned} & 0 \leq i+1 \wedge 0 \leq i-j \rightarrow \text{wlp}(c_0; c_1; c_2, (i+1, i-j) < (i'+1, i'-j'))[i/i', j/j'] \\ \Leftrightarrow & 0 < i \wedge 0 \leq i-j \wedge j < i \wedge a[j] \geq a[j+1] \rightarrow (i+1, i-(j+1)) < (i+1, i-j) \end{aligned}$$

The final basic path that we need to consider is:

```

1 invariant 0 <= i+1 && 0 <= i - j
2 decreases i+1, i-j
3 assume j >= i;           // c0
4 i := i - 1;              // c1
5 decreases i+1, i+1

```

This yields the verification condition:

$$\begin{aligned} & 0 \leq i+1 \wedge 0 \leq i-j \rightarrow \text{wlp}(c_0; c_1, (i+1, i-j) < (i'+1, i'+1))[i/i', j/j'] \\ \Leftrightarrow & 0 < i \wedge 0 \leq i-j \wedge j \geq i \rightarrow (i, i) < (i+1, i+1) \end{aligned}$$

This is obviously valid, so we conclude that the loop in **BubbleSort** terminates.

“Non-standard” Orderings Consider the following method, for which we would like to prove termination.

```

1 method M(x0: int, y0: int, z0: int)
2 {
3   var x, y, z := x0, y0, z0;
4   while x > 0 && y > 0 && z > 0
5   {
6     if y > x {
7       y := z;
8       x := *;
9       z := x - 1;
10    } else {
11      z := z - 1;
12      x := *;
13      y := x - 1;
14    }
15  }
16 }
```

This method contains a statement form that we haven’t seen before, namely non-deterministic assignment:

```
1 x := *;
```

The semantics of this statement are that x is assigned an arbitrary value from its domain; we can make no assumptions about what the value is, and when reasoning symbolically, we must treat it as a fresh constant.

$$\text{wlp}(x := *, Q) = Q[x/x'] \text{ where } x' \text{ is a fresh symbol not appearing in } Q$$

How do we prove that this method terminates. We can begin by observing that z definitely decreases in the **else** branch of the conditional statement. It also seems that z may decrease in the **if** branch, because we assign to it $x - 1$. Perhaps we should try using $\downarrow z$ as our ranking function.

```

1 method M(x0: int, y0: int, z0: int)
2 {
3   var x, y, z := x0, y0, z0;
4   while x > 0 && y > 0 && z > 0
5     decreases z
6   {
7     if y > x {
8       y := z;
9       x := *;
10      z := x - 1;
11    } else {
12      z := z - 1;
13      x := *;
14      y := x - 1;
15    }
16  }
17 }
```

Let’s first do a quick sanity check that this suffices in the branch that we expect it to work in,

considering the basic path corresponding to the `else`. We don't have a loop invariant, to the initial assertion in this path is simply *true*.

```

1 invariant true
2 decreases z
3 assume y <= x      // c0
4 z := z - 1;        // c1
5 x := *;            // c2
6 y := x - 1;        // c3
7 decreases z

```

Looking at the verification condition for this path,

$$\begin{aligned}
& \text{wlp}(c0; c1; c2; c3, z < z')[z/z'] \\
\Leftrightarrow & \text{wlp}(c0; c1; c2, z < z')[z/z'] \\
\Leftrightarrow & \text{wlp}(c0; c1, z < z')[z/z'] \\
\Leftrightarrow & \text{wlp}(c0, z - 1 < z')[z/z'] \\
\Leftrightarrow & (y \leq x \rightarrow z - 1 < z)
\end{aligned}$$

This is valid, but we also need to make sure that z is bounded from below to ensure well-foundedness. We simply need $0 \leq z$ in the antecedent of this condition, which we can obtain by adding the appropriate loop invariant.

```

1 invariant 0 <= z

```

Now let's take a look at the other branch.

```

1 invariant 0 <= z
2 decreases z
3 assume y > x      // c0
4 y := z;           // c1
5 x := *;           // c2
6 z := x - 1;       // c3
7 decreases z

```

This gives us the verification condition:

$$\begin{aligned}
& 0 \leq z \rightarrow \text{wlp}(c0; c1; c2; c3, z < z')[z/z'] \\
\Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0; c1; c2, x - 1 < z')[z/z'] \\
\Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0; c1, x'' - 1 < z')[z/z'] \\
\Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0, x'' - 1 < z')[z/z'] \\
\Leftrightarrow & (0 \leq z \wedge y > x) \rightarrow x'' - 1 < z
\end{aligned}$$

This is not valid, as x'' can take a value greater than z . We'll need to refine the verification condition to show that it decreases on this path.

Using a bit of intuition, we observe that a lexicographic ranking function might be a good choice, as we have a natural ranking function for the second branch, so perhaps we can find a function that decreases on the first branch and remains constant on the second. So, we're looking for some function f to insert into the lexicographic function:

$$(f(x, y, z), z)$$

We continue by observing that in the second branch, the condition $y > x$ remains unchanged: it is false whenever this branch is entered, and remains false because y is assigned $x - 1$ at the end of it. This is a Boolean-valued function, so if we assume an ordering on the Booleans, we may be able to use this in our first component. We're free to choose any ordering on the Booleans to build our lexicographic order, but according to our reasoning so far, it seems that we should use:

$$false < true$$

With this ordering over the Booleans in mind, our ranking function will be ordered lexicographically over pairs of Booleans and integers:

```
1 decreases y > x, z
```

To make our newly-constructed ordering completely explicit, let's write down a formula that characterizes whether a pair of tuples $(b_1, n_1) < (b_2, n_2)$.

$$(b_1, n_1) < (b_2, n_2) \\ \text{if and only if } (\neg b_1 \wedge b_2) \vee ((b_1 \leftrightarrow b_2) \wedge n_1 < n_2)$$

Extending this characterization to match the form of our ranking function, we have:

$$(y_1 > x_1, z_1) < (y_2 > x_2, z_2) \\ \text{if and only if } (\neg(y_1 > x_1) \wedge y_2 > x_2) \vee ((y_1 > x_1 \leftrightarrow y_2 > x_2) \wedge z_1 < z_2) \\ \Leftrightarrow (y_1 \leq x_1 \wedge y_2 > x_2) \vee ((y_1 > x_1 \leftrightarrow y_2 > x_2) \wedge z_1 < z_2)$$

Let's see if it works on the troublesome basic path from before.

```
1 invariant 0 <= z
2 decreases y > x, z
3 assume y > x           // c0
4 y := z;                 // c1
5 x := *;                 // c2
6 z := x - 1;             // c3
7 decreases y > x, z
```

We see that this gives us the verification condition:

$$\begin{aligned} & 0 \leq z \rightarrow \text{wlp}(c0; c1; c2; c3, (y > x, z) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0; c1; c2, (y > x, x - 1) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0; c1, (y > x'', x'' - 1) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0, (z > x'', x'' - 1) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & (0 \leq z \wedge y > x) \rightarrow (z > x'', x'' - 1) < (y > x, z) \end{aligned}$$

Is this valid? We can use the formula we derived earlier to decide by replacing the consequent of our verification condition with the requirements imposed by our ordering. This amounts to checking the validity of:

$$(0 \leq z \wedge y > x) \rightarrow ((z \leq x'' \wedge y > x) \vee ((z > x'' \leftrightarrow y > x) \wedge x'' - 1 < z))$$

This formula is fairly complex to reason about due to the multiple instance of disjunction (don't forget about the instance hiding in the biconditional). We know that because of the antecedent

condition $y > x$, the first component of the second tuple will be *true*. Thus, we can forget about trying to satisfy the first disjunct:

$$(z \leq x'' \wedge y > x)$$

We'll focus on the second disjunct, which according to our lexicographic ordering, forces the first component of the first tuple to either be *false*, i.e.,

$$z \leq x''$$

Or if it is not, then $x'' - 1$ must be less than z , i.e.,

$$z > x'' \wedge x'' - 1 < z$$

Putting all of this together, we arrive at a simpler formula whose validity implies that of the VC:

$$\begin{aligned} & (z \leq x'') \vee (z > x'' \wedge x'' - 1 < z) \\ \Leftrightarrow & z > x'' \rightarrow (z > x'' \wedge x'' - 1 < z) \\ \Leftrightarrow & z > x'' \rightarrow x'' - 1 < z \end{aligned}$$

This is valid, so the ranking function decreases on this path. Even though we don't expect the other basic path to pose a problem, let's go back and make sure that it decreases as it should.

```

1 invariant 0 <= z
2 decreases y > x, z
3 assume y <= x      // c0
4 z := z - 1;        // c1
5 x := *;             // c2
6 y := x - 1;        // c3
7 decreases y > x, z

```

With our new loop invariant and ranking function, we now have the verification condition:

$$\begin{aligned} & 0 \leq \text{zwlp}(c0; c1; c2; c3, (y > x, z) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0; c1; c2, (x - 1 > x, z) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0; c1, (x'' - 1 > x'', z) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & 0 \leq z \rightarrow \text{wlp}(c0, (x'' - 1 > x'', z - 1) < (y' > x', z'))[x/x', y/y', z/z'] \\ \Leftrightarrow & (0 \leq z \wedge y \leq x) \rightarrow (x'' - 1 > x'', z - 1) < (y > x, z) \end{aligned}$$

This formula is valid. On the left side of the inequality, we have:

$$(x'' - 1 > x'', z - 1) \text{ reduces to } (false, z - 1)$$

Keeping in mind the antecedent $y \leq x$, on the right side of the inequality we have:

$$(y > x, z) \text{ reduces to } (false, z)$$

So, the first component of the lexicographic comparison is an equality, because *false* = *false*. The second component compares $z - 1 < z$, which of course is *true*, making the verification condition valid. This completes the proof.